#48

```
> with( plots ):
> f := (x^3-x)/(x^6+1);
```

$$f := \frac{x^3 - x}{x^6 + 1} \qquad (1.1)$$

An antiderivative of f is

```
> int( f, x );
```

$$\frac{1}{6} \ln\left(x^4 - x^2 + 1\right) - \frac{1}{3} \ln\left(x^2 + 1\right) \qquad (1.2)$$

If you are curious how this answer could possibly be correct, notice that the denominator can be factored:

```
> factor( denom(f) );
```

$$\left(x^2 + 1\right) \left(x^4 - x^2 + 1\right) \qquad (1.3)$$

Then, the partial fraction decomposition of the integrand is

```
> convert( f, parfrac, x );
```

$$\frac{1}{3} \frac{x\left(2 x^2 - 1\right)}{x^4 - x^2 + 1} - \frac{2}{3} \frac{x}{x^2 + 1} \qquad (1.4)$$

Each term can now be integrated with a u-substitution with u=x^4-x^2+1 for the first term and u=x^2+1 for the second term. All-in-all, not so bad.

The problem asks us for the antiderivative with F(0)=0. The value of Maple's antiderivative, at x=0, is

```
> eval( (1.2), x=0 );
```

$$0 \qquad (1.5)$$

so this function satisfies F(0)=0:

```
> F := (1.2);
```

$$F := \frac{1}{6} \ln\left(x^4 - x^2 + 1\right) - \frac{1}{3} \ln\left(x^2 + 1\right) \qquad (1.6)$$

```
> fFplot := plot( [f,F], x=-4..4, linestyle=[dash,solid], legend=["y=f
  (x)","y=F(x)"], title="Plot for Section 7.6 #48" );
```

$$fFplot := PLOT(\ldots) \qquad (1.7)$$

```
> fFplot;
```



The extreme points of F occur when F'=f=0. The numerator of f is x^3-x^2, which is zero when x=-1, x=0, or x=1.

```
> pt := [x,F];
```

$$pt := \left[x, \frac{1}{6}\ln\left(x^4 - x^2 + 1\right) - \frac{1}{3}\ln\left(x^2+1\right)\right] \quad \textbf{(1.8)}$$

```
> eval( pt, x=-1 ); evalf( % );
```

$$\left[-1, -\frac{1}{3}\ln(2)\right]$$

$$[-1., -0.2310490602] \quad \textbf{(1.9)}$$

```
> eval( pt, x=0 );
```

$$[0, 0] \quad \textbf{(1.10)}$$

```
> eval( pt, x=1 ); evalf( % );
```

$$\left[1, -\frac{1}{3}\ln(2)\right]$$

$$[1., -0.2310490602] \quad \textbf{(1.11)}$$

The local (and global) maximum of F occurs at the point (0,0). There are two local extrema, which are also global minimums, at (1, -ln(2)/3) and at (-1, -ln(2)/3). The approximate value of ln(2)/3 is -0.231.

The plot suggets that there will be 4 inflection points. To find the location of these points, we need to find where F'' = f' = 0.

```
> df := diff( f, x );
```

(1.12)

$$df := \frac{3x^2-1}{x^6+1} - \frac{6(x^3-x)x^5}{(x^6+1)^2} \quad \textbf{(1.12)}$$

```
> simplify( (1.12) );
```

$$-\frac{3x^8-5x^6-3x^2+1}{(x^6+1)^2} \quad \textbf{(1.13)}$$

This suggests:

```
> solve( df=0, x );
```

$$\sqrt{RootOf(3\_Z^4-5\_Z^3-3\_Z+1, index=1)},\ -\sqrt{RootOf(3\_Z^4-5\_Z^3-3\_Z+1, index=1)},\ \sqrt{RootOf(3\_Z^4-5\_Z^3-3\_Z+1, index=2)},\ -\sqrt{RootOf(3\_Z^4-5\_Z^3-3\_Z+1, index=2)},\ \sqrt{RootOf(3\_Z^4-5\_Z^3-3\_Z+1, index=3)},\ -\sqrt{RootOf(3\_Z^4-5\_Z^3-3\_Z+1, index=3)},\ \sqrt{RootOf(3\_Z^4-5\_Z^3-3\_Z+1, index=4)},\ -\sqrt{RootOf(3\_Z^4-5\_Z^3-3\_Z+1, index=4)} \quad \textbf{(1.14)}$$

This is not useful. I can force Maple to give me a floating point answer if the problem includes floating point numbers. The easiest way to do this is to change 0 to 0.0 (or just 0.)

```
> solve( df=0.0, x );
```

$$-0.5452863556, 0.5452863556, -1.376934896, 1.376934896, -0.5028148771-0.7184239178\,I, 0.5028148771+0.7184239178\,I, -0.5028148771+0.7184239178\,I, 0.5028148771-0.7184239178\,I \quad \textbf{(1.15)}$$

Note that the final four solutions are complex-valued; these can be ignored. That leaves the first four (real-valued) values of x. These are about where would expect them to be. The actual inflection points are:

```
> eval( pt, x=(1.15)[1] );
```

$$[-0.5452863556, -0.1258322930] \quad \textbf{(1.16)}$$

```
> eval( pt, x=(1.15)[2] );
```

$$[0.5452863556, -0.1258322930] \quad \textbf{(1.17)}$$

```
> eval( pt, x=(1.15)[3] );
```

$$[-1.376934896, -0.1889775145] \quad \textbf{(1.18)}$$

```
> eval( pt, x=(1.15)[4] );
```

$$[1.376934896, -0.1889775145] \quad \textbf{(1.19)}$$

```
> ExtrPts := plot( [(1.9), (1.10), (1.11)], style=point, symbol=circle,
  color=red, symbolsize=16 );
```

$$ExtrPts := PLOT(...) \quad \textbf{(1.20)}$$

```
> InflPts := plot( [(1.16), (1.17), (1.18), (1.19)], style=point, symbol=box,
  color=green, symbolsize=16 );
```

$$InflPts := PLOT(...) \quad \textbf{(1.21)}$$

```
> display( [fFplot,ExtrPts,InflPts] );
```



Note that in terms of the derivative (f), the extreme points of F occur where f(x)=0 and the inflection points of F occur where f'(x) changes sign.

Additional Observation

If the value of Maple's antiderivative at x=0 was not zero, we would have to subtract this value from Maple's antiderivative. Let's see this on a different example:

```
> f := x^8*sin(x);
```

$$f := x^8 \sin(x) \quad \textbf{(1.1.1)}$$

```
> int( f, x );
```

$$-x^8 \cos(x) + 8\,x^7 \sin(x) + 56\,x^6 \cos(x) - 336\,x^5 \sin(x) - 1680\,x^4 \cos(x) + 6720\,x^3 \sin(x) + 20160\,x^2 \cos(x) - 40320 \cos(x) - 40320\,x \sin(x) \quad \textbf{(1.1.2)}$$

```
> F := (1.1.2) - eval( (1.1.2), x=0 );
```

$$F := -x^8 \cos(x) + 8\,x^7 \sin(x) + 56\,x^6 \cos(x) - 336\,x^5 \sin(x) - 1680\,x^4 \cos(x) + 6720\,x^3 \sin(x) + 20160\,x^2 \cos(x) - 40320 \cos(x) - 40320\,x \sin(x) + 40320 \quad \textbf{(1.1.3)}$$

If you look closely at this you will see that it's just the Fundamental Theorem of Calculus: F(x) = int(f, x=0..x);

```
> int( f, x=0..x );
```

$$-x^8 \cos(x) + 8\,x^7 \sin(x) + 56\,x^6 \cos(x) - 336\,x^5 \sin(x) - 1680\,x^4 \cos(x) + 6720\,x^3 \sin(x) + 20160\,x^2 \cos(x) - 40320 \cos(x) - 40320\,x \sin(x) + 40320 \quad \textbf{(1.1.4)}$$